

# HT67Fx0 Data EEPROM 之使用方法與注意事項

文件編碼：HA0254T

## 簡介

HT67Fx0 系列都有內部 EEPROM Data Memory 功能，EEPROM 資料暫存器容量為 64x8~256x8，用戶可以根據不同應用需求選擇合適 EEPROM 大小之 MCU。由於映射方式與程式儲存器和資料儲存器不同，因此不能像其他類型的儲存器一樣尋址。使用 Bank 0 中的一個位址和資料暫存器以及 Bank 1 中的一個控制暫存器，可以實現 EEPROM 的位元組讀寫操作。本文以 HT67F50 為母體，說明 EEPROM Data Memory 的使用方法。

## Theory of Operation

HT67Fx0 系列有三個暫存器控制內部 EEPROM 資料儲存器的操作。位址暫存器 EEA、資料暫存器 EED 及控制暫存器 EEC。其中 EEA 與 EED 位於 BANK0 中，它們能像其他特殊功能暫存器一樣直接被訪問。而 EEC 位於 BANK1 之中，不能被直接訪問，僅能通過 MP1 和 IAR1 進行間接讀取或寫入。由於 EEC 控制暫存器位於 Bank 1 中的[40H]，在 EEC 暫存器上的任何操作被執行前，MP1 必須先設為[40H]，BP 被設為[01H]。以下為 EEPROM 讀取操作或寫入操作的步驟：

- 設定所需要讀取/寫入資料的位址值 (EEA 暫存器的值)。
- 設定所需要寫入資料的值 (EED 暫存器的值)。
- 設定 BP.0 位元，選中資料儲存區 BANK1。
- 設定 MP1 為[40H]，指針指向 EEPROM 的控制暫存器 EEC。
- 使能讀取/寫入模式 (設定 RDEN/WREN 位元)。
- 開始讀取/寫入週期 (設定 RD/WR 位元)。
- 等待讀取/寫入週期結束。
- 讀取/寫入週期結束後，系統會自動清 RD 或 WR 為“0”，如果是寫入週期結束，這時其中斷請求旗標位元 DEF 會設定為“1” (HT67F50 目前讀取操作沒有中斷)。資料在其他讀取或寫入操作執行前將一直保留在 EED 暫存器中。應用程式可以通過查詢 RD 或 WR 位元以確定資料是否有效地被讀取或寫入。

## 程式說明

以下做了兩個範例以詳細說明 EEPROM 的相關讀取/寫入操作，其中範例 1 是以查尋相關讀取/寫入旗標位元的方式完成對 EEPROM 的讀取/寫入，主要說明如何設定位址暫存器 EEA、資料暫存器 EED 和控制暫存器 EEC 來實現 EEPROM 的讀取/寫入；範例 2 主要說明如何設定 EEPROM 相關中斷暫存器來實現寫入資料週期結束後進入中斷服務程式，以中斷的方式完成對 EEPROM 的寫入。

### 範例<1>：以查尋旗標位元的方式進行讀取/寫入EEPROM

；程式名：EEPROM 讀取/寫入 (eeprom\_rw.asm)

；目的：將資料寫入 EEPROM 一個位元組後再讀取出來相比較

；EEPROM 中斷：Disable

；Option：WDT disable

；程式清單：

```

INCLUDE HT67F50.INC
;-----
DATA .SECTION 'DATA'
EEPROM_ADRES DB ? ;user defined address
EEPROM_DATA DB ? ;user defined data
;-----
CODE .SECTION 'CODE'
    ORG 00H
    JMP MAIN
    ORG 04H ; External Interrupt 0 entry address
    RETI
    ORG 08H ; External Interrupt 1 entry address
    RETI
    ORG 0CH ; Comparator 0 Interrupt entry address
    RETI
    ORG 010H ; Comparator 1 Interrupt entry address
    RETI
    ORG 014H ; Multi-function Interrupt 0 entry address
    RETI
    ORG 018H ; Multi-function Interrupt 1 entry address
    RETI
    ORG 01CH ; A/D Converter Interrupt entry address
    RETI
    ORG 020H ; Multi-function Interrupt 2 entry address
    RETI
    ORG 024H ; Time base 0 Interrupt entry address
    RETI
    ORG 028H ; Time base 1 Interrupt entry address
    RETI
    ORG 02CH ; Multi-function Interrupt 3 entry address RETI
;-----
    ORG 30H
MAIN:
    CLR PAC ; PA set as output mode
    SET PAPU
    MOV A,0AH
    MOV WDTC,A ; close WDT
    CLR EMI ; Close Global interrupt
    SET HLCLK ; System clock selection for HXT
    CLR ACERL
    
```

```

        CLR CPOC
        CLR CP1C          ; Set as the I/O mode
;-----
;EEPROM WRITE OPERATION
;-----
WRITE_EEPROM:
    MOV A,10H
    MOV EEPROM_ADRES,A    ; Set write data address
    MOV A,055H
    MOV EEPROM_DATA,A     ; Set write data value
;-----
    MO  A,EEPROM_ADRES
    MOV EEA,A
    MOV A,EEPROM_DATA
    MOV EED,A
    SET BP.0              ; setup Bank Pointer
    MOV A,40H             ; setup memory pointer MP1
    MOV MP1,A            ; MP1 points to EEC register
    SET IAR1.3            ; set WREN bit, enable write operations
    SET IAR1.2            ; start Write Cycle - set WR bit
WRITE_WAIT:
    SZ  IAR1.2            ; check for write cycle end
    JMP WRITE_WAIT
;-----
    CLR IAR1              ; disable EEPROM read/write
    CLR BP
;-----
;EEPROM READ OPERATION
;-----
READ_EEPROM:
    MOV A,10H
    MOV EEPROM_ADRES,A    ; Set write data address
;-----
    MOV A,EEPROM_ADRES
    MOV EEA,A
    SET BP.0              ; setup Bank Pointer
    MOV A,40H             ; setup memory pointer MP1
    MOV MP1,A            ; MP1 points to EEC register
    SET IAR1.1            ; set RDEN bit, enable read operations
    SET IAR1.0            ; start Read Cycle - set RD bit
READ_WAIT:
    SZ  IAR1.0            ; check for read cycle end
    JMP READ_WAIT
    MOV A,EED              ; move read data to register
;-----
    XOR A,EEPROM_DATA     ; Read out value compared to the value written
    SZ  ACC                ; Determine the consistency of read data and write data
    JMP FAIL
    MOV A,01H
    XORM A,PA              ; If the same then PA0 output waveform
    JMP WRITE_EEPROM
;-----
FAIL:
    MOV A,02H
    XORM A,PA              ; If the inconsistency then PA1 output waveform
    JMP FAIL
    
```

### 程式說明

本範例主要介紹如何通過查詢旗標，以及正確設定相關暫存器來實現對 EEPROM 的讀取/寫入操作，在範例上只是簡單的循環寫入和讀取 EEPROM 的一個位元組的內容，其執行過程是對 EEPROM 的[10H]寫入 55H 後，再從 EEPROM 的[10H]讀取出來與寫入的值相比較，如果正確即 PA0 輸出波形，反之 PA1 輸出波形。通過示波器觀察 2 個 I/O 埠的輸出就可以判斷對 EEPROM 的讀取/寫入操作是否正常。

用戶使用時，只要呼叫下面讀取/寫入操作的副程式模組（完成一個指定位址的讀取和寫入），即可完成單位元組的讀取/寫入操作，如需要讀取/寫入多個位元組的內容時，只需在呼叫此副程式的同時修改所需要寫入和讀取時的位址變量 EEPROM\_ADRES 和資料變量 EEPROM\_DATA 就可以進行多位元組操作。讀取/寫入操作副程式如下：

讀取操作模組：

```

READ_EEPROM_LOOP:
    MOV A,EEPROM_ADRES
    MOV EEA,A
    SET BP.0           ; setup Bank Pointer
    MOV A,40H          ; setup memory pointer MP1
    MOV MP1,A          ; MP1 points to EEC register
    SET IAR1.1         ; set RDEN bit, enable read operations
    SET IAR1.0         ; start Read Cycle - set RD bit
READ_WAIT:
    SZ IAR1.0          ; check for read cycle end
    JMP READ_WAIT
    CLR IAR             ; disable EEPROM read/write
    CLR BP
    RET
    
```

寫操作模塊：

```

WRITE_EEPROM_LOOP:
    MOV A,EEPROM_ADRES
    MOV EEA,A
    MOV A,EEPROM_DATA
    MOV EED,A
    SET BP.0           ; setup Bank Pointer
    MOV A,40H          ; setup memory pointer MP1
    MOV MP1,A          ; MP1 points to EEC register
    SET IAR1.3         ; set WREN bit, enable write operations
    SET IAR1.2         ; start Write Cycle - set WR bit
WRITE_WAIT:
    SZ IAR1.2          ; check for write cycle end
    JMP WRITE_WAIT
    CLR IAR1           ; disable EEPROM read/write
    CLR BP
    RET
    
```

**範例< 2>：以中斷的方式完成對EEPROM的寫入操作**

; 程式名：EEPROM 寫入模式中斷 (eeprom\_r\_int.asm)  
 ; 目的：完成對 EEPROM 寫入操作後進入終端服務程式。  
 ; EEPROM 中斷：Enable  
 ; Option：WDT Disable  
 ; 程式清單：

```

INCLUDE HT67F50.INC
;-----
DATA .SECTION 'DATA'
EEPROM_ADRES DB ? ; user defined address
EEPROM_DATA DB ? ; user defined data
EEPROM_INT_FLAG DBIT ; Flag of interrupt
;-----
CODE .SECTION 'CODE'
ORG 00H
JMP MAIN
ORG 04H ; External interrupt 0 entry address
RETI
ORG 08H ; External interrupt 1 entry address
RETI
ORG 0CH ; Comparator 0 Interrupt entry address
RETI
ORG 010H ; Comparator 1 Interrupt entry address
RETI
ORG 014H ; Multi-function Interrupt 0 entry address
RETI
ORG 018H ; Multi-function Interrupt 1 entry address
RETI
ORG 01CH ; A/D Interrupt entry address
RETI
ORG 020H ; Multi-function Interrupt 2 entry address
RETI
ORG 024H ; Time base 0 Interrupt entry address
RETI
ORG ; Time base 1 Interrupt entry address
RETI
ORG 02CH ; Multi-function Interrupt 3 entry address
JMP EEPROM_INTERRUPT
;-----
ORG 30H
MAIN:
CLR PAC ; PA set as output mode
SET PAPU
SET HLCLK ; select system clock : HXT
CLR CMP0C ; set as I/O mode
CLR CMP1C ; set as I/O mode
CLR ACERL ; set as I/O mode
MOV A,01H
MOV INTC0,A ; enable EMI
MOV A,08H
MOV INTC2,A ; Enable Multi-Function Interrupt
MOV A,02H
MOV MFI3,A ; Enable EEPROM Interrupt
CLR EEPROM_INT_FLAG ; Initial interrupt flag
;-----
; EEPROM WRITE OPERATION
;-----
    
```

```

WRITE_EEPROM:
    MOV A,20H
    MOV EEPROM_ADRES,A      ; Set the address of data to be written
    MOV A,055H
    MOV EEPROM_DATA,A      ; Set the value of data to be written
;-----
WRITE_EEPROM_LOOP:
    MOV A,EEPROM_ADRES      ; user defined address
    MOV EEA,A
    MOV A,EEPROM_DATA      ; user defined data
    MOV EED,A
    SET BP.0                ; setup Bank Pointer
    MOV A,40H               ; setup memory pointer MP1
    MOV MP1,A              ; MP1 points to EEC register
    SET IAR1.3              ; set WREN bit, enable write operations
    SET IAR1.2              ; start Write Cycle - set WR bit
WRITE_WAIT:
    SNZ EEPROM_INT_FLAG    ; check for write cycle end
    JMP WRITE_WAIT
    CLR IAR1                ; disable EEPROM read/write
    CLR BP
    CLR EEPROM_INT_FLAG
    JMP WRITE_EEPROM
;-----
;EEPROM Interrupt Service Route
;-----
    ORG 0A0H
EEPROM_INTERRUPT:
    SET EEPROM_INT_FLAG
    MOV A,01H
    XORM A,PA               ; PA0 output waveform
    CLR MFI3.5              ; clr DEF
    RETI
    
```

### 程式說明

本範例程式主要說明如何設定 EEPROM 中斷有關的暫存器，在對 EEPROM 的一個寫入週期結束後產生中斷請求，並能進入中斷服務程式。與 EEPROM 中斷有關的暫存器有三個：總中斷允許位元 EMI[INTC0.0]、多功能中斷允許位元 MF3E[INTC2.3]和 EEPROM 中斷允許位元 DEE[MFI3.1]。要想能夠進入 EEPROM 中斷，這三個中斷允許位元必須先設定“1”。一但 EEPROM 一個寫入週期結束後，中斷請求旗標 DEF 會設定為“1”，在有關 EEPROM 中斷都使能的情況下即進入中斷服務程式。在中斷產生後，多功能中斷請求旗標 MF3F 位元會自動清除為“0”，而 EEPROM 中斷請求旗標 DEF 位元則不會自動清除為“0”，需要用軟體手動清除為“0”。本範例的執行過程是循環對 EEPROM 其中的一個位址[20H]寫入 55H 的資料，並在資料寫入週期結束後進入中斷服務，在中斷服務程式中設定一個旗標，用於判斷每寫完一次資料後是否有進入中斷服務程式，若有進入中斷則 PA0 有波形輸出。

## 注意事項

要執行對 EEPROM 的讀取/寫入操作時，必須按照先將 RDEN/WREN 設定為高位元，再將 RD/WR 設定為高位元的順序，才能開始讀取/寫入週期。

在同一條指令中 WREN、WR、RDEN 和 RD 不能同時設定為“1”。WR 和 RD 不能同時設定為“1”。

由於控制暫存器 EEC 位於 BANK1 中，不能被直接訪問，僅能通過 MP1 和 IAR1 進行間接讀取或寫入，在 EEC 暫存器上的任何操作被執行前，BP 必須先被設為[01H]，MP1 必須先設為[40H]。

中斷產生後，EEPROM 中斷請求旗標 DEF 位元不會自動清除為“0”，這時需要軟體來手動清除為“0”，以便下一次的中斷響應。

在正常程式操作中確保控制暫存器 EEC 中的寫入使能位元被清除，這樣能防止不正確的寫入操作。

## 版本記錄

版本：V1.10

修改人員：馬靈

修改日期：2012 年 1 月 9 日

修改內容：

- 更新文中 "EEA、EED、WDTC、MFI3、ACERL、CP0C、CP1C" 暫存器及 "DEF、DEE、MF3E" 位元的名稱。
- 更新程式附檔中 "EEA、EED、WDTC、MFI3、ACERL、CP0C、CP1C" 暫存器及 "DEF、DEE、MF3E" 位元的名稱。